

Introduction:

In this lab you will learn how to make fractals using NetLogo. Fractals are figures that are self-similar on all scales. This means that if the figure is repeatedly magnified the new figure looks similar to the original one. One way of generating a fractal is to take a basic pattern out of some elements (like a square or a rectangle) and then replace each of those elements with copies of the original pattern but shrunk to the appropriate size. This step is then repeated for as many steps as desired. Since this is a recursive processes, a computer programming language like NetLogo is an ideal doing this.

Sierpinski's Gasket:

In this example we will use a square as our basic element and we will build a triangular figure out of three squares. This is the generating pattern. In the next step we replace each square in the generating pattern with a reduced copy of the original pattern. Subsequent steps repeat this process. The starting square and the first three steps are shown. The resulting fractal is called Sierpinski's Gasket.



We can do this in Netlogo by starting with a single square turtle oriented with heading zero. The location of the turtle is the center of the square. At each step a turtle first reduces its size by one half. It then hatches two new turtles. One of which moves down and to the left by half the length of its side. The other moves down and to the right half the length of its side. The original turtle moves forward by half the length of its side. Make sure you understand how this process generates the pattern in the first step. This process is repeated as many times as needed (in practice 7 times is enough). When the last step is completed you ask all the turtles to stamp their shape and then die.

The following procedure implements the process described above:

```
to gasket
  cct 1 [set color green
        set shape "square"
        set size 100 ]
  repeat number [
    ask turtles [
      set size size / 2
      hatch 1 [ bk size / 2 lt 90 fd size / 2 rt 90]
      hatch 1 [ bk size / 2 rt 90 fd size / 2 lt 90]
      fd size / 2
    ]
  ]
  ask turtles [stamp die]
end
```

Read the procedure carefully and make sure you understand what each line does. If there is a command that you do not recognize look it up in the Primitives Directory.

Procedure:


Open up NetLogo. Now change the size of the view by clicking the edit icon on the view panel. Change both max-pxcor and max-pycor to 99. Also change the patch size to 3.0. Add a clear-all button to the interface. Now type in the gasket procedure in the procedures tab. Notice that you will need to assign values for the variable `number` using a slider on the interface. Make this slider range between 0 and 8 (any higher may cause NetLogo to crash). Add a `gasket` button to the interface and try out the procedure at various levels. Save your program using the naming convention:

`LastName_FirstName_Computer_Lab_07_Fractals.nlogo`

Extensions:

Try out the following extensions. When appropriate write new procedures and make a new button for each. Add sliders for any new variables you introduce:

- Change the shape of your basic element. You can use some of the standard shapes like “circle”, “arrow”, or “butterfly”, or you can use the Shapes Editor on the Tools menu to create your own shape. Make your shape rotatable.
- Change the colors of the turtles by a small amount each time a new turtle hatches: Colors change shade in small increments of 1 and they change to another color in the rainbow in increments of 10. So for example if the color is red the command `set color color + 10` changes the color to orange but `set color color + 1` changes it to a lighter shade of red.
- You can change the fractal dramatically by rotating one, or all the turtles by some angle at each step. For example, adding the command `rt 45` at the end of each hatch command and after the `fd step / 2` command generates the pattern on the right after 7 steps. Make your own modifications (you could rotate each turtle by a different amount or only rotate one).
- Make a generating pattern where the squares are placed in different places.
- Make a generating pattern that involves more than three squares (use extra hatch commands.)
- Make a generating pattern where some of the squares are smaller than others (change the size inside one or more of the hatch commands).



Making a movie:

NetLogo has a nice feature that will allow you to create an animation of the different views you get when changing a variable. For example, suppose you want to change the angle you rotate each of the elements in the basic Sierpinski pattern. You could do this by adding a global variable called `angle` (the first line of your procedure tab would be `globals [angle]`). Then modify the `gasket` procedure as follows:

```
to gasket
  cct 1 [set color green
        set shape "square"
        set size 100 ]
  repeat number [
    ask turtles [
      set size size / 2
      hatch 1 [ bk size / 2 lt 90 fd size / 2 rt 90 rt angle]
      hatch 1 [ bk size / 2 rt 90 fd size / 2 lt 90 rt angle]
      fd size / 2 rt angle
    ]
  ]
end
```

```
    ]  
    ask turtles [stamp die]  
end
```

Now we will use the movie making features of NetLogo to capture the view of the fractal for different angles ranging from 0 up to 360, in steps of 5 degrees.

First we create a new procedure. Lets call the procedure `make-a-movie`. We will save the movie as a “.mov” file. Lets call the movie `sierpinski_dance.mov`. The first frame of the movie will be with `angle` set to 0. Then we draw the fractal with the `gasket` procedure, grab the view and then increase the angle by 5 degrees and repeat. When we have repeated 72 times we close the movie. Here is the sequence of commands that do this.

```
to make-a-movie  
  set angle 0  
  movie-start "sierpinski_dance.mov"  
  repeat 72 [  
    clear-drawing  
    gasket  
    movie-grab-view  
    set angle angle + 5  
  ]  
  movie-close  
end
```


Add a button to the interface to make your movie. Note: the movie is saved in the same folder that you have saved your Netlogo file. If for some reason you have to halt your `make-a-movie` procedure in the middle, you will need to close the movie manually by typing `movie-close` in the command window. Try changing other parameters in your movie, like the initial color, or the number of step, or the initial location of the fractal. You will need to add a global variable for each parameter you wish to change.

Now that you have finished the lab, continue to the assignment below.

Fractals Assignment:

1. The first two steps of three different fractal mappings are given below. Use NetLogo to generate the seventh step. Some of the fractals are similar to the Sierpinski one, with the squares in the same location, but some or all the squares rotated. Others have the squares located in different locations and rotated. You will need to find out the locations and the rotation angles of each square. When you have figured out how to generate the fractal export the view of the 7th step as a “.png” image (right click on the NetLogo view panel or select File - Export – View on the menu). Name the picture according to the question number. Note: before starting use the *Shapes Editor* to modify the square shape so that it is rotatable, otherwise you will not be able to obtain the images in 1(b).

a)



b)




c)




2. Two fractals are given below. By examining the images determine what the generating pattern is and use NetLogo to check your answer. Export the 2nd step of each fractal as an image.

a)



b)



3. Create your own fractal. Use 3 or more squares in a generating pattern of your choice. I encourage you to consider using different colours, shapes, angles, locations and sizes of your basic elements. Export the 2nd and 7th order version of your fractal as images.
4. Using your fractal, or a version of one of the other fractals given in this assignment to generate a movie where you change one or more of the variables (angle of rotation, position, size, colour) in small steps. Name the movie:
Lastname_Firstname_FractalsQ4.mov)
5. To complete your assignment import all your images from questions 1-3 into a word document using the naming convention Lastname_Firstname_fractals.doc. Make sure you put the images in the correct order in the word document. You will now have three files to submit on Moodle: The NetLogofile, the movie from question 4, and the word documents with your answers to questions 1-3. This assignment is due on Tuesday, February 21st at 1:00pm.